
9.5 Roots of Polynomials

Here we present a few methods for finding roots of polynomials. These will
serve for most practical problems involving polynomials of low-to-moderate degree
or for well-conditioned polynomials of higher degree. Not as well appreciated as it
ought to be is the fact that some polynomials are exceedingly ill-conditioned. The
tiniest changes in a polynomial’s coefficients can, in the worst case, send its roots
sprawling all over the complex plane. (An infamous example due to Wilkinson is
detailed by Acton [1].)

Recall that a polynomial of degree n will have n roots. The roots can be real
or complex, and they might not be distinct. If the coefficients of the polynomial are
real, then complex roots will occur in pairs that are conjugate, i.e., if x1 = a + bi
is a root then x2 = a − bi will also be a root. When the coefficients are complex,
the complex roots need not be related.

Multiple roots, or closely spaced roots, produce the most difficulty for numerical
algorithms (see Figure 9.5.1). For example, P (x) = (x− a)2 has a double real root
at x = a. However, we cannot bracket the root by the usual technique of identifying
neighborhoods where the function changes sign, nor will slope-following methods
such as Newton-Raphson work well, because both the function and its derivative
vanish at a multiple root. Newton-Raphson may work, but slowly, since large
roundoff errors can occur. When a root is known in advance to be multiple, then
special methods of attack are readily devised. Problems arise when (as is generally
the case) we do not know in advance what pathology a root will display.

Deflation of Polynomials

When seeking several or all roots of a polynomial, the total effort can be
significantly reduced by the use of deflation. As each root r is found, the polynomial
is factored into a product involving the root and a reduced polynomial of degree
one less than the original, i.e., P (x) = (x − r)Q(x). Since the roots of Q are
exactly the remaining roots of P , the effort of finding additional roots decreases,
because we work with polynomials of lower and lower degree as we find successive
roots. Even more important, with deflation we can avoid the blunder of having our
iterative method converge twice to the same (nonmultiple) root instead of separately
to two different roots.

Deflation, which amounts to synthetic division, is a simple operation that acts
on the array of polynomial coefficients. The concise code for synthetic division by a
monomial factor was given in §5.3 above. You can deflate complex roots either by
converting that code to complex data type, or else — in the case of a polynomial with
real coefficients but possibly complex roots — by deflating by a quadratic factor,

[x− (a+ ib)] [x− (a− ib)] = x2 − 2ax+ (a2 + b2) (9.5.1)

(a)

x x

(b)

f (x) f (x)

Figure 9.5.1. (a) Linear, quadratic, and cubic behavior at the roots of polynomials. Only under high
magnification (b) does it become apparent that the cubic has one, not three, roots, and that the quadratic
has two roots rather than none.

The routine poldiv in §5.3 can be used to divide the polynomial by this factor.
Deflation must, however, be utilized with care. Because each new root is known

with only finite accuracy, errors creep into the determination of the coefficients of
the successively deflated polynomial. Consequently, the roots can become more and
more inaccurate. It matters a lot whether the inaccuracy creeps in stably (plus or
minus a few multiples of the machine precision at each stage) or unstably (erosion of
successive significant figures until the results become meaningless). Which behavior
occurs depends on just how the root is divided out. Forward deflation, where the
new polynomial coefficients are computed in the order from the highest power of x
down to the constant term, was illustrated in §5.3. This turns out to be stable if the
root of smallest absolute value is divided out at each stage. Alternatively, one can do
backward deflation, where new coefficients are computed in order from the constant
term up to the coefficient of the highest power of x. This is stable if the remaining
root of largest absolute value is divided out at each stage.

A polynomial whose coefficients are interchanged “end-to-end,” so that the
constant becomes the highest coefficient, etc., has its roots mapped into their
reciprocals. (Proof: Divide the whole polynomial by its highest power xn and
rewrite it as a polynomial in 1/x.) The algorithm for backward deflation is therefore
virtually identical to that of forward deflation, except that the original coefficients are
taken in reverse order and the reciprocal of the deflating root is used. Since we will
use forward deflation below, we leave to you the exercise of writing a concise coding
for backward deflation (as in §5.3). For more on the stability of deflation, consult [2].

To minimize the impact of increasing errors (even stable ones) when using
deflation, it is advisable to treat roots of the successively deflated polynomials as
only tentative roots of the original polynomial. One then polishes these tentative roots
by taking them as initial guesses that are to be re-solved for, using the nondeflated
original polynomialP . Again you must beware lest two deflated roots are inaccurate
enough that, under polishing, they both converge to the same undeflated root; in that
case you gain a spurious root-multiplicity and lose a distinct root. This is detectable,

since you can compare each polished root for equality to previous ones from distinct
tentative roots. When it happens, you are advised to deflate the polynomial just
once (and for this root only), then again polish the tentative root, or to use Maehly’s
procedure (see equation 9.5.29 below).

Below we say more about techniques for polishing real and complex-conjugate
tentative roots. First, let’s get back to overall strategy.

There are two schools of thought about how to proceed when faced with a
polynomial of real coefficients. One school says to go after the easiest quarry, the
real, distinct roots, by the same kinds of methods that we have discussed in previous
sections for general functions, i.e., trial-and-error bracketing followed by a safe
Newton-Raphson as in rtsafe. Sometimes you are only interested in real roots, in
which case the strategy is complete. Otherwise, you then go after quadratic factors
of the form (9.5.1) by any of a variety of methods. One such is Bairstow’s method,
which we will discuss below in the context of root polishing. Another is Muller’s
method, which we here briefly discuss.

Muller’s Method

Muller’s method generalizes the secant method, but uses quadratic interpolation
among three points instead of linear interpolation between two. Solving for the
zeros of the quadratic allows the method to find complex pairs of roots. Given three
previous guesses for the root xi−2, xi−1, xi, and the values of the polynomial P (x)
at those points, the next approximation xi+1 is produced by the following formulas,

q ≡ xi − xi−1

xi−1 − xi−2

A ≡ qP (xi) − q(1 + q)P (xi−1) + q2P (xi−2)

B ≡ (2q + 1)P (xi)− (1 + q)2P (xi−1) + q2P (xi−2)

C ≡ (1 + q)P (xi)

(9.5.2)

followed by

xi+1 = xi − (xi − xi−1)

[
2C

B ±
√
B2 − 4AC

]
(9.5.3)

where the sign in the denominator is chosen to make its absolute value or modulus
as large as possible. You can start the iterations with any three values of x that you
like, e.g., three equally spaced values on the real axis. Note that you must allow
for the possibility of a complex denominator, and subsequent complex arithmetic,
in implementing the method.

Muller’s method is sometimes also used for finding complex zeros of analytic
functions (not just polynomials) in the complex plane, for example in the IMSL
routine ZANLY [3].

Laguerre’s Method

The second school regarding overall strategy happens to be the one to which
we belong. That school advises you to use one of a very small number of methods
that will converge (though with greater or lesser efficiency) to all types of roots:
real, complex, single, or multiple. Use such a method to get tentative values for all
n roots of your nth degree polynomial. Then go back and polish them as you desire.

Laguerre’s method is by far the most straightforward of these general, complex
methods. It does require complex arithmetic, even while converging to real roots;
however, for polynomials with all real roots, it is guaranteed to converge to a
root from any starting point. For polynomials with some complex roots, little is
theoretically proved about the method’s convergence. Much empirical experience,
however, suggests that nonconvergence is extremely unusual, and, further, can almost
always be fixed by a simple scheme to break a nonconverging limit cycle. (This is
implemented in our routine, below.) An example of a polynomial that requires this
cycle-breaking scheme is one of high degree (>∼ 20), with all its roots just outside of
the complex unit circle, approximately equally spaced around it. When the method
converges on a simple complex zero, it is known that its convergence is third order.

In some instances the complex arithmetic in the Laguerre method is no
disadvantage, since the polynomial itself may have complex coefficients.

To motivate (although not rigorously derive) the Laguerre formulas we can note
the following relations between the polynomial and its roots and derivatives

Pn(x) = (x− x1)(x− x2) . . . (x − xn) (9.5.4)

ln |Pn(x)| = ln |x− x1|+ ln |x− x2|+ . . .+ ln |x− xn| (9.5.5)

d ln |Pn(x)|
dx

= +
1

x− x1
+

1

x− x2
+ . . .+

1

x− xn
=
P ′n
Pn
≡ G (9.5.6)

−d
2 ln |Pn(x)|

dx2
= +

1

(x− x1)2
+

1

(x− x2)2
+ . . .+

1

(x− xn)2

=

[
P ′n
Pn

]2

− P ′′n
Pn
≡ H (9.5.7)

Starting from these relations, the Laguerre formulas make what Acton [1] nicely calls
“a rather drastic set of assumptions”: The root x1 that we seek is assumed to be
located some distance a from our current guess x, while all other roots are assumed
to be located at a distance b

x− x1 = a ; x− xi = b i = 2, 3, . . . , n (9.5.8)

Then we can express (9.5.6), (9.5.7) as

1

a
+
n− 1

b
= G (9.5.9)

1

a2
+
n− 1

b2
= H (9.5.10)

which yields as the solution for a

a =
n

G±
√

(n− 1)(nH −G2)
(9.5.11)

where the sign should be taken to yield the largest magnitude for the denominator.
Since the factor inside the square root can be negative, a can be complex. (A more
rigorous justification of equation 9.5.11 is in [4].)

The method operates iteratively: For a trial value x, a is calculated by equation
(9.5.11). Then x − a becomes the next trial value. This continues until a is
sufficiently small.

The following routine implements the Laguerre method to find one root of a
given polynomial of degree m, whose coefficients can be complex. As usual, the first
coefficient a(1) is the constant term, while a(m+1) is the coefficient of the highest
power of x. The routine implements a simplified version of an elegant stopping
criterion due to Adams [5], which neatly balances the desire to achieve full machine
accuracy, on the one hand, with the danger of iterating forever in the presence of
roundoff error, on the other.

SUBROUTINE laguer(a,m,x,its)
INTEGER m,its,MAXIT,MR,MT
REAL EPSS
COMPLEX a(m+1),x
PARAMETER (EPSS=2.e-7,MR=8,MT=10,MAXIT=MT*MR)

Given the degree m and the complex coefficients a(1:m+1) of the polynomial
∑m+1
i=1 a(i)xi−1,

and given a complex value x, this routine improves x by Laguerre’s method until it con-
verges, within the achievable roundoff limit, to a root of the given polynomial. The number
of iterations taken is returned as its.
Parameters: EPSS is the estimated fractional roundoff error. We try to break (rare) limit
cycles with MR different fractional values, once every MT steps, for MAXIT total allowed
iterations.

INTEGER iter,j
REAL abx,abp,abm,err,frac(MR)
COMPLEX dx,x1,b,d,f,g,h,sq,gp,gm,g2
SAVE frac
DATA frac /.5,.25,.75,.13,.38,.62,.88,1./ Fractions used to break a limit cycle.
do 12 iter=1,MAXIT Loop over iterations up to allowed maximum.

its=iter
b=a(m+1)
err=abs(b)
d=cmplx(0.,0.)
f=cmplx(0.,0.)
abx=abs(x)
do 11 j=m,1,-1 Efficient computation of the polynomial and its first

two derivatives.f=x*f+d
d=x*d+b
b=x*b+a(j)
err=abs(b)+abx*err

enddo 11

err=EPSS*err Estimate of roundoff error in evaluating polynomial.
if(abs(b).le.err) then We are on the root.

return
else The generic case: use Laguerre’s formula.

g=d/b
g2=g*g
h=g2-2.*f/b
sq=sqrt((m-1)*(m*h-g2))
gp=g+sq
gm=g-sq
abp=abs(gp)
abm=abs(gm)
if(abp.lt.abm) gp=gm
if (max(abp,abm).gt.0.) then

dx=m/gp

else
dx=exp(cmplx(log(1.+abx),float(iter)))

endif
endif
x1=x-dx
if(x.eq.x1)return Converged.
if (mod(iter,MT).ne.0) then

x=x1
else Every so often we take a fractional step, to break any

limit cycle (itself a rare occurrence).x=x-dx*frac(iter/MT)
endif

enddo 12

pause ’too many iterations in laguer’ Very unusual — can occur only for complex roots.
return Try a different starting guess for the root.
END

Here is a driver routine that calls laguer in succession for each root, performs
the deflation, optionally polishes the roots by the same Laguerre method — if you
are not going to polish in some other way — and finally sorts the roots by their real
parts. (We will use this routine in Chapter 13.)

SUBROUTINE zroots(a,m,roots,polish)
INTEGER m,MAXM
REAL EPS
COMPLEX a(m+1),roots(m)
LOGICAL polish
PARAMETER (EPS=1.e-6,MAXM=101) A small number and maximum anticipated value of m+1.

C USES laguer

Given the degree m and the complex coefficients a(1:m+1) of the polynomial
∑m+1
i=1 a(i)xi−1,

this routine successively calls laguer and finds all m complex roots. The logical variable
polish should be input as .true. if polishing (also by Laguerre’s method) is desired,
.false. if the roots will be subsequently polished by other means.

INTEGER i,j,jj,its
COMPLEX ad(MAXM),x,b,c
do 11 j=1,m+1 Copy of coefficients for successive deflation.

ad(j)=a(j)
enddo 11

do 13 j=m,1,-1 Loop over each root to be found.
x=cmplx(0.,0.) Start at zero to favor convergence to smallest remaining root.
call laguer(ad,j,x,its) Find the root.
if(abs(aimag(x)).le.2.*EPS**2*abs(real(x))) x=cmplx(real(x),0.)
roots(j)=x
b=ad(j+1) Forward deflation.
do 12 jj=j,1,-1

c=ad(jj)
ad(jj)=b
b=x*b+c

enddo 12

enddo 13

if (polish) then
do 14 j=1,m Polish the roots using the undeflated coefficients.

call laguer(a,m,roots(j),its)
enddo 14

endif
do 16 j=2,m Sort roots by their real parts by straight insertion.

x=roots(j)
do 15 i=j-1,1,-1

if(real(roots(i)).le.real(x))goto 10
roots(i+1)=roots(i)

enddo 15

i=0

10 roots(i+1)=x
enddo 16

return
END

Eigenvalue Methods

The eigenvalues of a matrix A are the roots of the “characteristic polynomial”
P (x) = det[A − xI]. However, as we will see in Chapter 11, root-finding is not
generally an efficient way to find eigenvalues. Turning matters around, we can
use the more efficient eigenvalue methods that are discussed in Chapter 11 to find
the roots of arbitrary polynomials. You can easily verify (see, e.g., [6]) that the
characteristic polynomial of the special m ×m companion matrix

A =


− am
am+1

−am−1
am+1

· · · − a2
am+1

− a1
am+1

1 0 · · · 0 0
0 1 · · · 0 0
...

...
0 0 · · · 1 0

 (9.5.12)

is equivalent to the general polynomial

P (x) =
m+1∑
i=1

aix
i−1 (9.5.13)

If the coefficients ai are real, rather than complex, then the eigenvalues of A can be
found using the routines balanc and hqr in §§11.5–11.6 (see discussion there). This
method, implemented in the routine zrhqr following, is typically about a factor 2
slower than zroots (above). However, for some classes of polynomials, it is a more
robust technique, largely because of the fairly sophisticated convergence methods
embodied in hqr. If your polynomial has real coefficients, and you are having
trouble with zroots, then zrhqr is a recommended alternative.

SUBROUTINE zrhqr(a,m,rtr,rti)
INTEGER m,MAXM
REAL a(m+1),rtr(m),rti(m)
PARAMETER (MAXM=50)

C USES balanc,hqr

Find all the roots of a polynomial with real coefficients,
∑m+1
i=1 a(i)xi−1, given the degree

m and the coefficients a(1:m+1). The method is to construct an upper Hessenberg matrix
whose eigenvalues are the desired roots, and then use the routines balanc and hqr. The
real and imaginary parts of the roots are returned in rtr(1:m) and rti(1:m), respectively.

INTEGER j,k
REAL hess(MAXM,MAXM),xr,xi
if (m.gt.MAXM.or.a(m+1).eq.0.) pause ’bad args in zrhqr’
do 12 k=1,m Construct the matrix.

hess(1,k)=-a(m+1-k)/a(m+1)
do 11 j=2,m

hess(j,k)=0.
enddo 11

if (k.ne.m) hess(k+1,k)=1.

enddo 12

call balanc(hess,m,MAXM) Find its eigenvalues.
call hqr(hess,m,MAXM,rtr,rti)
do 14 j=2,m Sort roots by their real parts by straight insertion.

xr=rtr(j)
xi=rti(j)
do 13 k=j-1,1,-1

if(rtr(k).le.xr)goto 1
rtr(k+1)=rtr(k)
rti(k+1)=rti(k)

enddo 13

k=0
1 rtr(k+1)=xr

rti(k+1)=xi
enddo 14

return
END

Other Sure-Fire Techniques

The Jenkins-Traub method has become practically a standard in black-box
polynomial root-finders, e.g., in the IMSL library [3]. The method is too complicated
to discuss here, but is detailed, with references to the primary literature, in [4].

The Lehmer-Schur algorithm is one of a class of methods that isolate roots in
the complex plane by generalizing the notion of one-dimensional bracketing. It is
possible to determine efficiently whether there are any polynomial roots within a
circle of given center and radius. From then on it is a matter of bookkeeping to
hunt down all the roots by a series of decisions regarding where to place new trial
circles. Consult [1] for an introduction.

Techniques for Root-Polishing

Newton-Raphson works very well for real roots once the neighborhood of
a root has been identified. The polynomial and its derivative can be efficiently
simultaneously evaluated as in §5.3. For a polynomial of degree n-1with coefficients
c(1)...c(n), the following segment of code embodies one cycle of Newton-
Raphson:

p=c(n)*x+c(n-1)
p1=c(n)
do 11 i=n-2,1,-1

p1=p+p1*x
p=c(i)+p*x

enddo 11

if (p1.eq.0.) pause ’derivative should not vanish’
x=x-p/p1

Once all real roots of a polynomial have been polished, one must polish the
complex roots, either directly, or by looking for quadratic factors.

Direct polishing by Newton-Raphson is straightforward for complex roots if the
above code is converted to complex data types. With real polynomial coefficients,
note that your starting guess (tentative root) must be off the real axis, otherwise
you will never get off that axis — and may get shot off to infinity by a minimum
or maximum of the polynomial.

For real polynomials, the alternative means of polishing complex roots (or, for that
matter, double real roots) is Bairstow’s method, which seeks quadratic factors. The advantage
of going after quadratic factors is that it avoids all complex arithmetic. Bairstow’s method
seeks a quadratic factor that embodies the two roots x = a ± ib, namely

x2 − 2ax + (a2 + b2) ≡ x2 +Bx +C (9.5.14)

In general if we divide a polynomial by a quadratic factor, there will be a linear remainder

P (x) = (x2 +Bx +C)Q(x) + Rx+ S. (9.5.15)

Given B and C, R and S can be readily found, by polynomial division (§5.3). We can
consider R and S to be adjustable functions of B and C, and they will be zero if the
quadratic factor is zero.

In the neighborhood of a root a first-order Taylor series expansion approximates the
variation of R,S with respect to small changes in B,C

R(B + δB,C + δC) ≈ R(B,C) +
∂R

∂B
δB +

∂R

∂C
δC (9.5.16)

S(B + δB,C + δC) ≈ S(B, C) +
∂S

∂B
δB +

∂S

∂C
δC (9.5.17)

To evaluate the partial derivatives, consider the derivative of (9.5.15) with respect toC. Since
P (x) is a fixed polynomial, it is independent of C, hence

0 = (x2 + Bx+C)
∂Q

∂C
+Q(x) +

∂R

∂C
x +

∂S

∂C
(9.5.18)

which can be rewritten as

−Q(x) = (x2 +Bx +C)
∂Q

∂C
+
∂R

∂C
x+

∂S

∂C
(9.5.19)

Similarly, P (x) is independent of B, so differentiating (9.5.15) with respect to B gives

−xQ(x) = (x2 +Bx+ C)
∂Q

∂B
+
∂R

∂B
x +

∂S

∂B
(9.5.20)

Now note that equation (9.5.19) matches equation (9.5.15) in form. Thus if we perform a
secondsynthetic division ofP (x), i.e., a division ofQ(x), yielding a remainderR1x+S1, then

∂R

∂C
= −R1

∂S

∂C
= −S1 (9.5.21)

To get the remaining partial derivatives, evaluate equation (9.5.20) at the two roots of the
quadratic, x+ and x−. Since

Q(x±) = R1x± + S1 (9.5.22)

we get

∂R

∂B
x+ +

∂S

∂B
= −x+(R1x+ + S1) (9.5.23)

∂R

∂B
x− +

∂S

∂B
= −x−(R1x− + S1) (9.5.24)

Solve these two equations for the partial derivatives, using

x+ + x− = −B x+x− = C (9.5.25)

and find
∂R

∂B
= BR1 − S1

∂S

∂B
= CR1 (9.5.26)

Bairstow’s method now consists of using Newton-Raphson in two dimensions (which is
actually the subject of the next section) to find a simultaneous zero of R and S. Synthetic
division is used twice per cycle to evaluate R,S and their partial derivatives with respect to
B,C. Like one-dimensional Newton-Raphson, the method works well in the vicinity of a root
pair (real or complex), but it can fail miserably when started at a random point. We therefore
recommend it only in the context of polishing tentative complex roots.

SUBROUTINE qroot(p,n,b,c,eps)
INTEGER n,NMAX,ITMAX
REAL b,c,eps,p(n),TINY
PARAMETER (NMAX=20,ITMAX=20,TINY=1.0e-6)

C USES poldiv
Given coefficients p(1:n) of a polynomial of degree n-1, and trial values for the coefficients
of a quadratic factor x*x+b*x+c, improve the solution until the coefficients b,c change
by less than eps. The routine poldiv §5.3 is used.
Parameters: At most NMAX coefficients, ITMAX iterations.

INTEGER iter
REAL delb,delc,div,r,rb,rc,s,sb,sc,d(3),q(NMAX),qq(NMAX),rem(NMAX)
d(3)=1.
do 11 iter=1,ITMAX

d(2)=b
d(1)=c
call poldiv(p,n,d,3,q,rem)
s=rem(1) First division r,s.
r=rem(2)
call poldiv(q,n-1,d,3,qq,rem)
sc=-rem(1) Second division partial r,s with respect to c.
rc=-rem(2)
sb=-c*rc
rb=sc-b*rc
div=1./(sb*rc-sc*rb) Solve 2x2 equation.
delb=(r*sc-s*rc)*div
delc=(-r*sb+s*rb)*div
b=b+delb
c=c+delc
if((abs(delb).le.eps*abs(b).or.abs(b).lt.TINY)

* .and.(abs(delc).le.eps*abs(c)
* .or.abs(c).lt.TINY)) return Coefficients converged.

enddo 11

pause ’too many iterations in qroot’
END

We have already remarked on the annoyance of having two tentative roots
collapse to one value under polishing. You are left not knowing whether your
polishing procedure has lost a root, or whether there is actually a double root,
which was split only by roundoff errors in your previous deflation. One solution
is deflate-and-repolish; but deflation is what we are trying to avoid at the polishing
stage. An alternative is Maehly’s procedure. Maehly pointed out that the derivative
of the reduced polynomial

Pj(x) ≡ P (x)

(x − x1) · · · (x− xj)
(9.5.27)

can be written as

P ′j(x) =
P ′(x)

(x− x1) · · · (x− xj)
− P (x)

(x− x1) · · · (x− xj)

j∑
i=1

(x− xi)−1 (9.5.28)

Hence one step of Newton-Raphson, taking a guess xk into a new guess xk+1,
can be written as

xk+1 = xk −
P (xk)

P ′(xk)− P (xk)
∑j
i=1(xk − xi)−1

(9.5.29)

This equation, if used with i ranging over the roots already polished, will prevent a
tentative root from spuriously hopping to another one’s true root. It is an example
of so-called zero suppression as an alternative to true deflation.

Muller’s method, which was described above, can also be useful at the
polishing stage.

9.6 Newton-Raphson Method for Nonlinear
Systems of Equations

We make an extreme, but wholly defensible, statement: There are no good, gen-
eral methods for solving systems of more than one nonlinear equation. Furthermore,
it is not hard to see why (very likely) there never will be any good, general methods:
Consider the case of two dimensions, where we want to solve simultaneously

f(x, y) = 0

g(x, y) = 0
(9.6.1)

The functions f and g are two arbitrary functions, each of which has zero
contour lines that divide the (x, y) plane into regions where their respective function
is positive or negative. These zero contour boundaries are of interest to us. The
solutions that we seek are those points (if any) that are common to the zero contours
of f and g (see Figure 9.6.1). Unfortunately, the functions f and g have, in general,
no relation to each other at all! There is nothing special about a common point from
either f’s point of view, or from g’s. In order to find all common points, which are
the solutions of our nonlinear equations, we will (in general) have to do neither more
nor less than map out the full zero contours of both functions. Note further that
the zero contours will (in general) consist of an unknown number of disjoint closed
curves. How can we ever hope to know when we have found all such disjoint pieces?

For problems in more than two dimensions, we need to find points mutually
common to N unrelated zero-contour hypersurfaces, each of dimension N − 1.


